Rules for integrands of the form $(a + b x^n)^p (c + d x^n)^q$

1: $\int (a + b x^n)^p (c + d x^n)^q \, dx$ when $b c - a d \neq 0 \land (p \mid q) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.3.1: If $b c - a d \neq 0 \land (p \mid q) \in \mathbb{Z}^+$, then

$$\int (a + b x^n)^p (c + d x^n)^q \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(a + b x^n)^p (c + d x^n)^q, x\right] dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n)^p*(c+d*x^n)^q,x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && IGtQ[p,0] && IGtQ[q,0]
```

2: $\int (a + b x^n)^p (c + d x^n)^q \, dx$ when $b c - a d \neq 0 \land (p \mid q) \in \mathbb{Z} \land n < 0$

Derivation: Algebraic expansion

Basis: If $p \in \mathbb{Z}$, then $(a + b x^n)^p == x^{n p} (b + a x^{-n})^p$

–

Rule 1.1.3.3.2: If $b c - a d \neq 0 \land (p \mid q) \in \mathbb{Z} \land n < 0$, then

$$\int (a + b x^n)^p (c + d x^n)^q \, dx \longrightarrow \int x^{n (p+q)} (b + a x^{-n})^p (d + c x^{-n})^q \, dx$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  Int[x^(n*(p+q))*(b+a*x^(-n))^p*(d+c*x^(-n))^q,x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && IntegersQ[p,q] && NegQ[n]
```

3: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $F[x] == -\text{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule 1.1.3.3.3: If $b\,c-a\,d\neq 0\,\wedge\,n\in\mathbb{Z}^-$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow\;-\text{Subst}\left[\int\frac{(a+b\,x^{-n})^p\,(c+d\,x^{-n})^q}{x^2}\,dx,\,x,\,\frac{1}{x}\right]$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n))^p*(c+d*x^(-n))^q/x^2,x],x,1/x] /;
FreeQ[{a,b,c,d,p,q},x] && NeQ[b*c-a*d,0] && ILtQ[n,0]
```

4: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g\in\mathbb{Z}^+$, then $F[x^n] == g\,\text{Subst}\left[x^{g-1}\,F[x^{g\,n}],\,x,\,x^{1/g}\right]\,\partial_x x^{1/g}$

Rule 1.1.3.3.4: If $b\,c-a\,d\neq 0\,\land\,n\in\mathbb{F}$, let $g=\text{Denominator}[n]$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow\;g\,\text{Subst}\left[\int x^{g-1}\,(a+b\,x^{g\,n})^p\,(c+d\,x^{g\,n})^q\,dx,\,x,\,x^{1/g}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g-1)*(a+b*x^(g*n))^p*(c+d*x^(g*n))^q,x],x,x^(1/g)]] /;
FreeQ[{a,b,c,d,p,q},x] && NeQ[b*c-a*d,0] && FractionQ[n]
```

5. $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c-a d \neq 0 \wedge n (p+q+1)+1 == 0$

1. $\int \frac{(a+b x^n)^p}{c+d x^n} \, dx$ when $b c-a d \neq 0 \wedge n p+1 == 0 \wedge n \in \mathbb{Z}$

1: $\int \frac{1}{(a+b x^3)^{1/3} (c+d x^3)} \, dx$ when $b c-a d \neq 0$

- Note: This rule for cubic binomials is optional, but leads to slightly simpler results than the following one.

- Rule 1.1.3.3.5.1.1: If $b c-a d \neq 0$, let $q \to \left(\frac{b c-a d}{c}\right)^{1/3}$, then

$$\int \frac{1}{(a+b x^3)^{1/3} (c+d x^3)} \, dx \longrightarrow \frac{\operatorname{ArcTan}\left[\frac{1+\frac{2 q x}{(a+b x^3)^{1/3}}}{\sqrt{3}}\right]}{\sqrt{3}\, c q}+\frac{\operatorname{Log}[c+d x^3]}{6 c q}-\frac{\operatorname{Log}\left[q x-(a+b x^3)^{1/3}\right]}{2 c q}$$

- Program code:

```
Int[1/((a_+b_.*x_^3)^(1/3)*(c_+d_.*x_^3)),x_Symbol] :=
  With[{q=Rt[(b*c-a*d)/c,3]},
  ArcTan[(1+(2*q*x)/(a+b*x^3)^(1/3))/Sqrt[3]]/(Sqrt[3]*c*q) + Log[c+d*x^3]/(6*c*q) - Log[q*x-(a+b*x^3)^(1/3)]/(2*c*q)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

2: $\int \frac{(a+b x^n)^p}{c+d x^n} \, dx$ when $b c-a d \neq 0 \wedge n p+1 == 0 \wedge n \in \mathbb{Z}$

- Derivation: Integration by substitution

- Basis: If $n \in \mathbb{Z}$, then $\frac{1}{(a+b x^n)^{1/n} (c+d x^n)} == \operatorname{Subst}\left[\frac{1}{c-(b c-a d) x^n}, x, \frac{x}{(a+b x^n)^{1/n}}\right] \partial_x \frac{x}{(a+b x^n)^{1/n}}$

- Rule 1.1.3.3.5.1.2: If $b c-a d \neq 0 \wedge n p+1 == 0 \wedge n \in \mathbb{Z}$, then

$$\int \frac{(a+b\,x^n)^p}{c+d\,x^n}\,dx \longrightarrow \text{Subst}\left[\int \frac{1}{c-(b\,c-a\,d)\,x^n}\,dx,\ x,\ \frac{x}{(a+b\,x^n)^{1/n}}\right]$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_/(c_+d_.*x_^n_),x_Symbol] :=
  Subst[Int[1/(c-(b*c-a*d)*x^n),x],x,x/(a+b*x^n)^(1/n)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[n*p+1,0] && IntegerQ[n]
```

2: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\wedge n\,(p+q+1)+1==0\wedge q>0\wedge p\neq -1$

Derivation: Binomial product recurrence 1 with $A=1$, $B=0$ and $n\,(p+q+1)+1==0$

–

Note: If this kool rules applies, it will also apply to the resulting integrands until p and q are reduced to the interval [-1,0).

Rule 1.1.3.3.5.2: If $b\,c-a\,d\neq 0\wedge n\,(p+q+1)+1==0\wedge q>0\wedge p\neq -1$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx \longrightarrow -\frac{x\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^q}{a\,n\,(p+1)}-\frac{c\,q}{a\,(p+1)}\int (a+b\,x^n)^{p+1}\,(c+d\,x^n)^{q-1}\,dx$$

–

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_.,x_Symbol] :=
  -x*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*n*(p+1)) -
  c*q/(a*(p+1))*Int[(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1),x] /;
FreeQ[{a,b,c,d,n,p},x] && NeQ[b*c-a*d,0] && EqQ[n*(p+q+1)+1,0] && GtQ[q,0] && NeQ[p,-1]
```

3: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\wedge n\,(p+q+1)+1==0\wedge p\in\mathbb{Z}^-$

Rule 1.1.3.3.5.3: If $b\,c-a\,d\neq 0\wedge n\,(p+q+1)+1==0\wedge p\in\mathbb{Z}^-$, then

$$\int (a + b x^n)^p (c + d x^n)^q \, dx \longrightarrow \frac{a^p x}{c^{p+1} (c + d x^n)^{1/n}} \text{Hypergeometric2F1}\left[\frac{1}{n}, -p, 1 + \frac{1}{n}, -\frac{(b c - a d) x^n}{a (c + d x^n)}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  a^p*x/(c^(p+1)*(c+d*x^n)^(1/n))*Hypergeometric2F1[1/n,-p,1+1/n,-(b*c-a*d)*x^n/(a*(c+d*x^n))] /;
FreeQ[{a,b,c,d,n,q},x] && NeQ[b*c-a*d,0] && EqQ[n*(p+q+1)+1,0] && ILtQ[p,0]
```

4: $\int (a + b x^n)^p (c + d x^n)^q \, dx$ when $b c - a d \neq 0 \land n (p + q + 1) + 1 == 0$

Rule 1.1.3.3.5.4: If $b c - a d \neq 0 \land n (p + q + 1) + 1 == 0$, then

$$\int (a + b x^n)^p (c + d x^n)^q \, dx \longrightarrow \frac{x (a + b x^n)^p}{c \left(\frac{c (a+b x^n)}{a (c+d x^n)}\right)^p (c + d x^n)^{\frac{1}{n}+p}} \text{Hypergeometric2F1}\left[\frac{1}{n}, -p, 1 + \frac{1}{n}, -\frac{(b c - a d) x^n}{a (c + d x^n)}\right]$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  x*(a+b*x^n)^p/(c*(c*(a+b*x^n)/(a*(c+d*x^n)))^p*(c+d*x^n)^(1/n+p))*
    Hypergeometric2F1[1/n,-p,1+1/n,-(b*c-a*d)*x^n/(a*(c+d*x^n))] /;
FreeQ[{a,b,c,d,n,p,q},x] && NeQ[b*c-a*d,0] && EqQ[n*(p+q+1)+1,0]
```

6. $\int (a + b x^n)^p (c + d x^n)^q \, dx$ when $b c - a d \neq 0 \land n (p + q + 2) + 1 == 0$

1: $\int (a + b x^n)^p (c + d x^n)^q \, dx$ when $b c - a d \neq 0 \land n (p + q + 2) + 1 == 0 \land a d (p + 1) + b c (q + 1) == 0$

Derivation: Binomial product recurrence 2a with $A = 1$, $B = 0$ and $n (p + q + 2) + 1 == 0$

Rule 1.1.3.3.6.1: If $b c - a d \neq 0 \land n (p + q + 2) + 1 == 0 \land a d (p + 1) + b c (q + 1) == 0$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow \frac{x (a+b x^n)^{p+1} (c+d x^n)^{q+1}}{a c}$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  x*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*c) /;
FreeQ[{a,b,c,d,n,p,q},x] && NeQ[b*c-a*d,0] && EqQ[n*(p+q+2)+1,0] && EqQ[a*d*(p+1)+b*c*(q+1),0]
```

```
(* Int[(a1_+b1_.*x_^n2_.)^p_*(a2_+b2_.*x_^n2_.)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  x*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)*(c+d*x^n)^(q+1)/(a1*a2*c) /;
FreeQ[{a1,b1,a2,b2,c,d,n,p,q},x] && EqQ[n2,n/2] && EqQ[a2*b1+a1*b2,0] && EqQ[n*(p+q+2)+1,0] && EqQ[a1*a2*d*(p+1)+b1*b2*c*(q+1),0] *)
```

2: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land n (p+q+2)+1 = 0 \land p < -1$

Derivation: Binomial product recurrence 2a with $A = 1$, $B = 0$ and $n (p+q+2)+1 = 0$

Note: Note the resulting integrand is of the form $(a+b x^n)^p (c+d x^n)^q$ where $n (p+q+1)+1 = 0$.

Rule 1.1.3.3.6.2: If $b c - a d \neq 0 \land n (p+q+2)+1 = 0 \land p < -1$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow -\frac{b x (a+b x^n)^{p+1} (c+d x^n)^{q+1}}{a n (p+1) (b c - a d)} + \frac{b c + n (p+1) (b c - a d)}{a n (p+1) (b c - a d)} \int (a+b x^n)^{p+1} (c+d x^n)^q \, dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -b*x*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*n*(p+1)*(b*c-a*d)) +
  (b*c+n*(p+1)*(b*c-a*d))/(a*n*(p+1)*(b*c-a*d))*Int[(a+b*x^n)^(p+1)*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,n,q},x] && NeQ[b*c-a*d,0] && EqQ[n*(p+q+2)+1,0] && (LtQ[p,-1] || Not[LtQ[q,-1]]) && NeQ[p,-1]
```

7. $\int (a+b x^n)^p (c+d x^n) \, dx$ when $b c - a d \neq 0$

1: $\int (a+b x^n)^p (c+d x^n) \, dx$ when $b c - a d \neq 0 \land a d - b c (n (p+1)+1) == 0$

Derivation: Trinomial recurrence 2b with $c = 0$, $p = 0$ and $a d - b c (n (p+1)+1) == 0$

Rule 1.1.3.3.7.1: If $b c - a d \neq 0 \land a d - b c (n (p+1)+1) == 0$, then

$$\int (a+b x^n)^p (c+d x^n) \, dx \longrightarrow \frac{c x (a+b x^n)^{p+1}}{a}$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*x*(a+b*x^n)^(p+1)/a /;
FreeQ[{a,b,c,d,n,p},x] && NeQ[b*c-a*d,0] && EqQ[a*d-b*c*(n*(p+1)+1),0]
```

```
Int[(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  c*x*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(a1*a2) /;
FreeQ[{a1,b1,a2,b2,c,d,n,p},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && EqQ[a1*a2*d-b1*b2*c*(n*(p+1)+1),0]
```

2: $\int (a+b x^n)^p (c+d x^n) \, dx$ when $b c - a d \neq 0 \land p < -1$

Derivation: Trinomial recurrence 2b with $c = 0$ and $p = 0$

Rule 1.1.3.3.7.2: If $b c - a d \neq 0 \land p < -1$, then

$$\int (a+b x^n)^p (c+d x^n) \, dx \longrightarrow -\frac{(b c - a d) \, x \, (a+b x^n)^{p+1}}{a b n (p+1)} - \frac{a d - b c (n (p+1) + 1)}{a b n (p+1)} \int (a+b x^n)^{p+1} \, dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_),x_Symbol] :=
  -(b*c-a*d)*x*(a+b*x^n)^(p+1)/(a*b*n*(p+1)) -
  (a*d-b*c*(n*(p+1)+1))/(a*b*n*(p+1))*Int[(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,d,n,p},x] && NeQ[b*c-a*d,0] && (LtQ[p,-1] || ILtQ[1/n+p,0])
```

```
Int[(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  -(b1*b2*c-a1*a2*d)*x*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(a1*a2*b1*b2*n*(p+1)) -
  (a1*a2*d-b1*b2*c*(n*(p+1)+1))/(a1*a2*b1*b2*n*(p+1))*Int[(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1),x] /;
FreeQ[{a1,b1,a2,b2,c,d,n},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && (LtQ[p,-1] || ILtQ[1/n+p,0])
```

3: $\int \frac{c + d x^n}{a + b x^n} \, dx$ when $b c - a d \neq 0 \land n < 0$

Derivation: Algebraic expansion

Basis: $\frac{c+d x^n}{a+b x^n} = \frac{c}{a} - \frac{b c-a d}{a \left(b+a x^{-n}\right)}$

Rule 1.1.3.3.7.3: If $b c - a d \neq 0 \land n < 0$, then

$$\int \frac{c + d x^n}{a + b x^n} \, dx \longrightarrow \frac{c x}{a} - \frac{b c - a d}{a} \int \frac{1}{b + a x^{-n}} \, dx$$

Program code:

```
Int[(c_+d_.*x_^n_)/(a_+b_.*x_^n_),x_Symbol] :=
  c*x/a - (b*c-a*d)/a*Int[1/(b+a*x^(-n)),x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && LtQ[n,0]
```

4: $\int \left(a + b x^n\right)^p \left(c + d x^n\right) dx$ when $b c - a d \neq 0 \land n (p + 1) + 1 \neq 0$

Derivation: Trinomial recurrence 2b with $c = 0$ and $p = 0$ composed with binomial recurrence 1b with $p = 0$

Rule 1.1.3.3.7.4: If $b c - a d \neq 0 \land n (p + 1) + 1 \neq 0$, then

$$\int \left(a + b x^n\right)^p \left(c + d x^n\right) dx \longrightarrow \frac{d x \left(a + b x^n\right)^{p+1}}{b (n (p + 1) + 1)} - \frac{a d - b c (n (p + 1) + 1)}{b (n (p + 1) + 1)} \int \left(a + b x^n\right)^p dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_),x_Symbol] :=
  d*x*(a+b*x^n)^(p+1)/(b*(n*(p+1)+1)) -
  (a*d-b*c*(n*(p+1)+1))/(b*(n*(p+1)+1))*Int[(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && NeQ[n*(p+1)+1,0]
```

```
Int[(a1_+b1_.*x_^non2_.)^p_.*(a2_+b2_.*x_^non2_.)^p_.*(c_+d_.*x_^n_),x_Symbol] :=
  d*x*(a1+b1*x^(n/2))^(p+1)*(a2+b2*x^(n/2))^(p+1)/(b1*b2*(n*(p+1)+1)) -
  (a1*a2*d-b1*b2*c*(n*(p+1)+1))/(b1*b2*(n*(p+1)+1))*Int[(a1+b1*x^(n/2))^p*(a2+b2*x^(n/2))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,d,n,p},x] && EqQ[non2,n/2] && EqQ[a2*b1+a1*b2,0] && NeQ[n*(p+1)+1,0]
```

8: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^- \land p \geq -q$

Derivation: Algebraic expansion

Rule 1.1.3.3.8: If $b c - a d \neq 0 \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^- \land p \geq -q$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \rightarrow \int \text{PolynomialDivide}\left[(a+b x^n)^p,\ (c+d x^n)^{-q},\ x\right] dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  Int[PolynomialDivide[(a+b*x^n)^p,(c+d*x^n)^(-q),x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && IGtQ[p,0] && ILtQ[q,0] && GeQ[p,-q]
```

9. $\int \frac{(a+b\,x^n)^p}{c+d\,x^n}\,dx$ when $b\,c-a\,d\neq 0$

0: $\int \frac{(a+b\,x^n)^p}{c+d\,x^n}\,dx$ when $b\,c-a\,d\neq 0\ \wedge\ n\,(p-1)+1=0\ \wedge\ n\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\frac{(a+b\,z)^p}{c+d\,z}=\frac{b\,(a+b\,z)^{p-1}}{d}-\frac{(b\,c-a\,d)\,(a+b\,z)^{p-1}}{d\,(c+d\,z)}$

Rule 1.1.3.3.9.0: If $b\,c-a\,d\neq 0\ \wedge\ n\,(p-1)+1=0\ \wedge\ n\in\mathbb{Z}$, then

$$\int \frac{(a+b\,x^n)^p}{c+d\,x^n}\,dx\ \longrightarrow\ \frac{b}{d}\int (a+b\,x^n)^{p-1}\,dx-\frac{b\,c-a\,d}{d}\int \frac{(a+b\,x^n)^{p-1}}{c+d\,x^n}\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_/(c_+d_.*x_^n_),x_Symbol] :=
  b/d*Int[(a+b*x^n)^(p-1),x] - (b*c-a*d)/d*Int[(a+b*x^n)^(p-1)/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b*c-a*d,0] && EqQ[n*(p-1)+1,0] && IntegerQ[n]
```

1: $\int \frac{1}{(a+b x^n)(c+d x^n)} dx$ when $b c - a d \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b z)(c+d z)} == \frac{b}{(b c-a d)(a+b z)} - \frac{d}{(b c-a d)(c+d z)}$

Rule 1.1.3.3.9.1: If $b c - a d \neq 0$, then

$$\int \frac{1}{(a+b x^n)(c+d x^n)} dx \to \frac{b}{(b c-a d)} \int \frac{1}{a+b x^n} dx - \frac{d}{(b c-a d)} \int \frac{1}{c+d x^n} dx$$

Program code:

```
Int[1/((a_+b_.*x_^n_)*(c_+d_.*x_^n_)),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(a+b*x^n),x] - d/(b*c-a*d)*Int[1/(c+d*x^n),x] /;
FreeQ[{a,b,c,d,n},x] && NeqQ[b*c-a*d,0]
```

2. $\int \frac{(a+b x^2)^p}{c+d x^2} \, dx$ when $b c - a d \neq 0$

1. $\int \frac{1}{(a+b x^2)^{1/3} (c+d x^2)} \, dx$ when $b c - a d \neq 0 \wedge (b c + 3 a d == 0 \vee b c - 9 a d == 0)$

1. $\int \frac{1}{(a+b x^2)^{1/3} (c+d x^2)} \, dx$ when $b c - a d \neq 0 \wedge b c + 3 a d == 0$

1: $\int \frac{1}{(a+b x^2)^{1/3} (c+d x^2)} \, dx$ when $b c - a d \neq 0 \wedge b c + 3 a d == 0 \wedge \frac{b}{a} > 0$

Derivation: Integration by substitution

Basis: $F\left[(a+b x^2)^{1/3}, x^2\right] == \frac{3\sqrt{b x^2}}{2 b x} \operatorname{Subst}\left[\frac{x^2}{\sqrt{-a+x^3}} F\left[x, \frac{-a+x^3}{b}\right], x, (a+b x^2)^{1/3}\right] \partial_x (a+b x^2)^{1/3}$

Rule 1.1.3.3.9.2.1.1.1: If $b c - a d \neq 0 \wedge b c + 3 a d == 0 \wedge \frac{b}{a} > 0$, let $q \to \sqrt{\frac{b}{a}}$, then

$$\int \frac{1}{(a+b x^2)^{1/3} (c+d x^2)} \, dx \to \frac{3\sqrt{b x^2}}{2 x} \operatorname{Subst}\left[\int \frac{x}{\sqrt{-a+x^3} \, (b c - a d + d x^3)} \, dx, x, (a+b x^2)^{1/3}\right]$$

$$\to \frac{q \operatorname{ArcTanh}\left[\frac{\sqrt{3}}{q x}\right]}{2 \times 2^{2/3} \sqrt{3} \, a^{1/3} d} + \frac{q \operatorname{ArcTanh}\left[\frac{\sqrt{3} \left(a^{1/3} - 2^{1/3} (a+b x^2)^{1/3}\right)}{a^{1/3} q x}\right]}{2 \times 2^{2/3} \sqrt{3} \, a^{1/3} d} + \frac{q \operatorname{ArcTan}[q x]}{6 \times 2^{2/3} a^{1/3} d} - \frac{q \operatorname{ArcTan}\left[\frac{a^{1/3} q x}{a^{1/3} + 2^{1/3} (a+b x^2)^{1/3}}\right]}{2 \times 2^{2/3} a^{1/3} d}$$

Program code:

```
Int[1/((a_+b_.*x_^2)^(1/3)*(c_+d_.*x_^2)),x_Symbol] :=
  With[{q=Rt[b/a,2]},
  q*ArcTanh[Sqrt[3]/(q*x)]/(2*2^(2/3)*Sqrt[3]*a^(1/3)*d) +
  q*ArcTanh[Sqrt[3]*(a^(1/3)-2^(1/3)*(a+b*x^2)^(1/3))/(a^(1/3)*q*x)]/(2*2^(2/3)*Sqrt[3]*a^(1/3)*d) +
  q*ArcTan[q*x]/(6*2^(2/3)*a^(1/3)*d) -
   q*ArcTan[(a^(1/3)*q*x)/(a^(1/3)+2^(1/3)*(a+b*x^2)^(1/3))]/(2*2^(2/3)*a^(1/3)*d)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b*c+3*a*d,0] && PosQ[b/a]
```

2: $\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx$ when $b c - a d \neq 0 \land b c + 3 a d == 0 \land \frac{b}{a} \ngtr 0$

- Rule 1.1.3.3.9.2.1.1.2: If $b c - a d \neq 0 \land b c + 3 a d == 0 \land \frac{b}{a} \ngtr 0$, let $q \to \sqrt{-\frac{b}{a}}$, then

$$\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx \longrightarrow \frac{q \operatorname{ArcTan}\left[\frac{\sqrt{3}}{q x}\right]}{2 \times 2^{2/3} \sqrt{3} \, a^{1/3} d} + \frac{q \operatorname{ArcTan}\left[\frac{\sqrt{3} \left(a^{1/3}-2^{1/3} \left(a+b x^2\right)^{1/3}\right)}{a^{1/3} q x}\right]}{2 \times 2^{2/3} \sqrt{3} \, a^{1/3} d} - \frac{q \operatorname{ArcTanh}[q x]}{6 \times 2^{2/3} a^{1/3} d} + \frac{q \operatorname{ArcTanh}\left[\frac{a^{1/3} q x}{a^{1/3}+2^{1/3} \left(a+b x^2\right)^{1/3}}\right]}{2 \times 2^{2/3} a^{1/3} d}$$

- Program code:

```
Int[1/((a_+b_.*x_^2)^(1/3)*(c_+d_.*x_^2)),x_Symbol] :=
  With[{q=Rt[-b/a,2]},
  q*ArcTan[Sqrt[3]/(q*x)]/(2*2^(2/3)*Sqrt[3]*a^(1/3)*d) +
  q*ArcTan[Sqrt[3]*(a^(1/3)-2^(1/3)*(a+b*x^2)^(1/3))/(a^(1/3)*q*x)]/(2*2^(2/3)*Sqrt[3]*a^(1/3)*d) -
  q*ArcTanh[q*x]/(6*2^(2/3)*a^(1/3)*d) +
   q*ArcTanh[(a^(1/3)*q*x)/(a^(1/3)+2^(1/3)*(a+b*x^2)^(1/3))]/(2*2^(2/3)*a^(1/3)*d)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b*c+3*a*d,0] && NegQ[b/a]
```

2. $\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx$ when $b c - a d \neq 0 \land b c - 9 a d == 0$

1: $\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx$ when $b c - a d \neq 0 \land b c - 9 a d == 0 \land \frac{b}{a} > 0$

Rule 1.1.3.3.9.2.1.2.1.1: If $b c - a d \neq 0 \land b c - 9 a d == 0 \land \frac{b}{a} > 0$, let $q \to \sqrt{\frac{b}{a}}$, then

$$\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx \to -\frac{q \operatorname{ArcTan}\left[\frac{q x}{3}\right]}{12 a^{1/3} d} + \frac{q \operatorname{ArcTan}\left[\frac{a^{1/3}-\left(a+b x^2\right)^{1/3}}{a^{1/3} q x}\right]}{12 a^{1/3} d} - \frac{q \operatorname{ArcTan}\left[\frac{a^{1/3}+2\left(a+b x^2\right)^{1/3}}{a^{1/3} q x}\right]}{12 a^{1/3} d} - \frac{q \operatorname{ArcTanh}\left[\frac{\sqrt{3}\left(a^{1/3}-\left(a+b x^2\right)^{1/3}\right)}{a^{1/3} q x}\right]}{4 \sqrt{3} a^{1/3} d}$$

$$\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx \to \frac{q \operatorname{ArcTan}\left[\frac{q x}{3}\right]}{12 a^{1/3} d} + \frac{q \operatorname{ArcTan}\left[\frac{\left(a^{1/3}-\left(a+b x^2\right)^{1/3}\right)^2}{3 a^{2/3} q x}\right]}{12 a^{1/3} d} - \frac{q \operatorname{ArcTanh}\left[\frac{\sqrt{3}\left(a^{1/3}-\left(a+b x^2\right)^{1/3}\right)}{a^{1/3} q x}\right]}{4 \sqrt{3} a^{1/3} d}$$

Program code:

```
Int[1/((a_+b_.*x_^2)^(1/3)*(c_+d_.*x_^2)),x_Symbol] :=
  With[{q=Rt[b/a,2]},
  q*ArcTan[q*x/3]/(12*Rt[a,3]*d) +
  q*ArcTan[(Rt[a,3]-(a+b*x^2)^(1/3))^2/(3*Rt[a,3]^2*q*x)]/(12*Rt[a,3]*d) -
  q*ArcTanh[(Sqrt[3]*(Rt[a,3]-(a+b*x^2)^(1/3)))/(Rt[a,3]*q*x)]/(4*Sqrt[3]*Rt[a,3]*d)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b*c-9*a*d,0] && PosQ[b/a]
```

2: $\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx$ when $b c - a d \neq 0 \land b c - 9 a d == 0 \land \frac{b}{a} \not> 0$

Rule 1.1.3.3.9.2.1.2.1.1: If $b c - a d \neq 0 \land b c - 9 a d == 0 \land \frac{b}{a} \not> 0$, let $q \to \sqrt{-\frac{b}{a}}$, then

$$\int \frac{1}{\left(a+b x^2\right)^{1/3} \left(c+d x^2\right)} \, dx \to -\frac{q \operatorname{ArcTanh}\left[\frac{q x}{3}\right]}{12 a^{1/3} d} + \frac{q \operatorname{ArcTanh}\left[\frac{a^{1/3}-\left(a+b x^2\right)^{1/3}}{a^{1/3} q x}\right]}{12 a^{1/3} d} - \frac{q \operatorname{ArcTanh}\left[\frac{a^{1/3}+2\left(a+b x^2\right)^{1/3}}{a^{1/3} q x}\right]}{12 a^{1/3} d} - \frac{q \operatorname{ArcTan}\left[\frac{\sqrt{3}\left(a^{1/3}-\left(a+b x^2\right)^{1/3}\right)}{a^{1/3} q x}\right]}{4 \sqrt{3} a^{1/3} d}$$

$$\int \frac{1}{\left(a+b\,x^2\right)^{1/3}\left(c+d\,x^2\right)}\,dx \longrightarrow -\frac{q\,\text{ArcTanh}\left[\frac{q\,x}{3}\right]}{12\,a^{1/3}\,d}+\frac{q\,\text{ArcTanh}\left[\frac{\left(a^{1/3}-\left(a+b\,x^2\right)^{1/3}\right)^2}{3\,a^{2/3}\,q\,x}\right]}{12\,a^{1/3}\,d}-\frac{q\,\text{ArcTan}\left[\frac{\sqrt{3}\,\left(a^{1/3}-\left(a+b\,x^2\right)^{1/3}\right)}{a^{1/3}\,q\,x}\right]}{4\,\sqrt{3}\,a^{1/3}\,d}$$

Program code:

```
Int[1/((a_+b_.*x_^2)^(1/3)*(c_+d_.*x_^2)),x_Symbol] :=
  With[{q=Rt[-b/a,2]},
  -q*ArcTanh[q*x/3]/(12*Rt[a,3]*d) +
  q*ArcTanh[(Rt[a,3]-(a+b*x^2)^(1/3))^2/(3*Rt[a,3]^2*q*x)]/(12*Rt[a,3]*d) -
  q*ArcTan[(Sqrt[3]*(Rt[a,3]-(a+b*x^2)^(1/3)))/(Rt[a,3]*q*x)]/(4*Sqrt[3]*Rt[a,3]*d)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b*c-9*a*d,0] && NegQ[b/a]
```

2: $\int \frac{\left(a+b\,x^2\right)^{2/3}}{c+d\,x^2}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,b\,c+3\,a\,d==0$

Derivation: Algebraic expansion

Basis: $\frac{\left(a+b\,x^2\right)^{2/3}}{c+d\,x^2}==\frac{b}{d\,\left(a+b\,x^2\right)^{1/3}}-\frac{b\,c-a\,d}{d\,\left(a+b\,x^2\right)^{1/3}\,\left(c+d\,x^2\right)}$

Rule 1.1.3.3.9.2.2: If $b\,c-a\,d\neq 0\,\wedge\,b\,c+3\,a\,d==0$, then

$$\int \frac{\left(a+b\,x^2\right)^{2/3}}{c+d\,x^2}\,dx \longrightarrow \frac{b}{d}\int \frac{1}{\left(a+b\,x^2\right)^{1/3}}\,dx-\frac{b\,c-a\,d}{d}\int \frac{1}{\left(a+b\,x^2\right)^{1/3}\,\left(c+d\,x^2\right)}\,dx$$

Program code:

```
Int[(a_+b_.*x_^2)^(2/3)/(c_+d_.*x_^2),x_Symbol] :=
  b/d*Int[1/(a+b*x^2)^(1/3),x] - (b*c-a*d)/d*Int[1/((a+b*x^2)^(1/3)*(c+d*x^2)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b*c+3*a*d,0]
```

3. $\int \frac{1}{\left(a+b\,x^2\right)^{1/4}\,\left(c+d\,x^2\right)}\,dx$ when $b\,c-a\,d\neq 0$

1. $\int \frac{1}{\left(a+b x^2\right)^{1/4}\left(c+d x^2\right)} dx$ when $b c-2 a d=0$

1: $\int \frac{1}{\left(a+b x^2\right)^{1/4}\left(c+d x^2\right)} dx$ when $b c-2 a d=0 \land \frac{b^2}{a}>0$

- Reference: Eneström index number E688 in The Euler Archive

- Rule 1.1.3.3.9.2.3.1.1: If $b c-2 a d=0 \land \frac{b^2}{a}>0$, let $q \to \left(\frac{b^2}{a}\right)^{1/4}$, then

$$\int \frac{1}{\left(a+b x^2\right)^{1/4}\left(c+d x^2\right)} dx \to -\frac{b}{2 a d q} \text{ArcTan}\left[\frac{b+q^2 \sqrt{a+b x^2}}{q^3 x \left(a+b x^2\right)^{1/4}}\right]-\frac{b}{2 a d q} \text{ArcTanh}\left[\frac{b-q^2 \sqrt{a+b x^2}}{q^3 x \left(a+b x^2\right)^{1/4}}\right]$$

- Program code:

```
Int[1/((a_+b_.*x_^2)^(1/4)*(c_+d_.*x_^2)),x_Symbol] :=
  With[{q=Rt[b^2/a,4]},
  -b/(2*a*d*q)*ArcTan[(b+q^2*Sqrt[a+b*x^2])/(q^3*x*(a+b*x^2)^(1/4))] -
   b/(2*a*d*q)*ArcTanh[(b-q^2*Sqrt[a+b*x^2])/(q^3*x*(a+b*x^2)^(1/4))]] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && PosQ[b^2/a]
```

2: $\int \frac{1}{\left(a+b x^2\right)^{1/4}\left(c+d x^2\right)} dx$ when $b c-2 a d=0 \land \frac{b^2}{a} \ngtr 0$

Reference: Eneström index number E688 in The Euler Archive

- Derivation: Integration by substitution

- Basis: If $b c-2 a d=0$, then $\frac{1}{\left(a+b x^2\right)^{1/4}\left(c+d x^2\right)}=\frac{2 b}{d} \text{Subst}\left[\frac{1}{4 a+b^2 x^4}, x, \frac{x}{\left(a+b x^2\right)^{1/4}}\right] \partial_x \frac{x}{\left(a+b x^2\right)^{1/4}}$

Rule 1.1.3.3.9.2.3.1.2: If $b c-2 a d=0 \land \frac{b^2}{a} \ngtr 0$, let $q \to \left(-\frac{b^2}{a}\right)^{1/4}$, then

$$\int \frac{1}{\left(a+b x^2\right)^{1/4}\left(c+d x^2\right)} dx \to \frac{2 b}{d} \text{Subst}\left[\int \frac{1}{4 a+b^2 x^4} dx, x, \frac{x}{\left(a+b x^2\right)^{1/4}}\right]$$

$$\longrightarrow \frac{b}{2\sqrt{2}\, a\, d\, q}\,\text{ArcTan}\left[\frac{q\, x}{\sqrt{2}\,\left(a+b\, x^2\right)^{1/4}}\right]+\frac{b}{2\sqrt{2}\, a\, d\, q}\,\text{ArcTanh}\left[\frac{q\, x}{\sqrt{2}\,\left(a+b\, x^2\right)^{1/4}}\right]$$

–

Program code:

```
Int[1/((a_+b_.*x_^2)^(1/4)*(c_+d_.*x_^2)),x_Symbol] :=
  With[{q=Rt[-b^2/a,4]},
  b/(2*Sqrt[2]*a*d*q)*ArcTan[q*x/(Sqrt[2]*(a+b*x^2)^(1/4))] +
  b/(2*Sqrt[2]*a*d*q)*ArcTanh[q*x/(Sqrt[2]*(a+b*x^2)^(1/4))]] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && NegQ[b^2/a]
```

x: $\int \frac{1}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} \, dx$ when $b c - 2 a d == 0 \land \frac{b^2}{a} \not> 0$

– Reference: Eneström index number E688 in The Euler Archive

Derivation: Integration by substitution

– Basis: If $b c - 2 a d == 0$, then $\frac{1}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} == \frac{2 b}{d} \text{Subst}\left[\frac{1}{4 a+b^2 x^4}, x, \frac{x}{\left(a+b x^2\right)^{1/4}}\right] \partial_x \frac{x}{\left(a+b x^2\right)^{1/4}}$

– Note: Although this antiderivative is real and continuous when the integrand is real, it is unnecessarily discontinuous when the integrand is not real.

– Rule 1.1.3.3.9.2.3.1.2: If $b c - 2 a d == 0 \land \frac{b^2}{a} \not> 0$, let $q \to \left(-\frac{b^2}{a}\right)^{1/4}$, then

$$\int \frac{1}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} \, dx \to \frac{2 b}{d} \text{Subst}\left[\int \frac{1}{4 a+b^2 x^4} \, dx, x, \frac{x}{\left(a+b x^2\right)^{1/4}}\right]$$

$$\to \frac{b}{2 \sqrt{2} \, a d q} \text{ArcTan}\left[\frac{q x}{\sqrt{2} \left(a+b x^2\right)^{1/4}}\right] + \frac{b}{4 \sqrt{2} \, a d q} \text{Log}\left[\frac{\sqrt{2} \, q x + 2 \left(a+b x^2\right)^{1/4}}{\sqrt{2} \, q x - 2 \left(a+b x^2\right)^{1/4}}\right]$$

– Program code:

```
(* Int[1/((a_+b_.*x_^2)^(1/4)*(c_+d_.*x_^2)),x_Symbol] :=
  With[{q=Rt[-b^2/a,4]},
  b/(2*Sqrt[2]*a*d*q)*ArcTan[q*x/(Sqrt[2]*(a+b*x^2)^(1/4))] +
  b/(4*Sqrt[2]*a*d*q)*Log[(Sqrt[2]*q*x+2*(a+b*x^2)^(1/4))/(Sqrt[2]*q*x-2*(a+b*x^2)^(1/4))]] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0] && NegQ[b^2/a] *)
```

2: $\int \frac{1}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} \, dx$ when $b c-a d \neq 0$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{-\frac{b x^2}{a}}}{x} == 0$

- Basis: $\frac{x}{\sqrt{-\frac{b x^2}{a}} \left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} == 2 \, \text{Subst}\left[\frac{x^2}{\sqrt{1-\frac{x^4}{a}} \left(b c-a d+d x^4\right)}, x, \left(a+b x^2\right)^{1/4}\right] \partial_x \left(a+b x^2\right)^{1/4}$

- Rule 1.1.3.3.9.2.3.2: If $b c-a d \neq 0$, then

$$\int \frac{1}{\left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} \, dx \rightarrow \frac{\sqrt{-\frac{b x^2}{a}}}{x} \int \frac{x}{\sqrt{-\frac{b x^2}{a}} \left(a+b x^2\right)^{1/4} \left(c+d x^2\right)} \, dx \rightarrow$$

$$\frac{2 \sqrt{-\frac{b x^2}{a}}}{x} \, \text{Subst}\left[\int \frac{x^2}{\sqrt{1-\frac{x^4}{a}} \left(b c-a d+d x^4\right)} \, dx, x, \left(a+b x^2\right)^{1/4}\right]$$

- Program code:

```
Int[1/((a_+b_.*x_^2)^(1/4)*(c_+d_.*x_^2)),x_Symbol] :=
  2*Sqrt[-b*x^2/a]/x*Subst[Int[x^2/(Sqrt[1-x^4/a]*(b*c-a*d+d*x^4)),x],x,(a+b*x^2)^(1/4)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

4. $\int \frac{1}{\left(a+b\,x^2\right)^{3/4}\,\left(c+d\,x^2\right)}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{1}{\left(a+b\,x^2\right)^{3/4}\,\left(c+d\,x^2\right)}\,dx$ when $b\,c-2\,a\,d==0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\left(a+b\,x^2\right)^{3/4}\,\left(c+d\,x^2\right)}==\frac{1}{c\,\left(a+b\,x^2\right)^{3/4}}-\frac{d\,x^2}{c\,\left(a+b\,x^2\right)^{3/4}\,\left(c+d\,x^2\right)}$

Note: There are terminal rules for $\int \frac{x^2}{\left(a+b\,x^2\right)^{3/4}\,\left(c+d\,x^2\right)}\,dx$ when $b\,c-2\,a\,d==0$.

Rule 1.1.3.3.9.2.4.1: If $b\,c-2\,a\,d==0$, then

$$\int \frac{1}{\left(a+b\,x^2\right)^{3/4}\,\left(c+d\,x^2\right)}\,dx\ \longrightarrow\ \frac{1}{c}\int \frac{1}{\left(a+b\,x^2\right)^{3/4}}\,dx-\frac{d}{c}\int \frac{x^2}{\left(a+b\,x^2\right)^{3/4}\,\left(c+d\,x^2\right)}\,dx$$

Program code:

```
Int[1/((a_+b_.*x_^2)^(3/4)*(c_+d_.*x_^2)),x_Symbol] :=
  1/c*Int[1/(a+b*x^2)^(3/4),x] - d/c*Int[x^2/((a+b*x^2)^(3/4)*(c+d*x^2)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-2*a*d,0]
```

2: $\int \frac{1}{\left(a+b x^2\right)^{3/4} \left(c+d x^2\right)} \, dx$ when $b c - a d \neq 0$

- Derivation: Piecewise constant extranction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{-\frac{b x^2}{a}}}{x} = 0$

- Basis: $x F\left[x^2\right] = \frac{1}{2} \text{Subst}\left[F[x], x, x^2\right] \partial_x x^2$

- Rule 1.1.3.3.9.2.4.2: If $b c - a d \neq 0$, then

$$\int \frac{1}{\left(a+b x^2\right)^{3/4} \left(c+d x^2\right)} \, dx \to \frac{\sqrt{-\frac{b x^2}{a}}}{x} \int \frac{x}{\sqrt{-\frac{b x^2}{a}} \left(a+b x^2\right)^{3/4} \left(c+d x^2\right)} \, dx \to$$

$$\frac{\sqrt{-\frac{b x^2}{a}}}{2 x} \text{Subst}\left[\int \frac{1}{\sqrt{-\frac{b x}{a}} (a+b x)^{3/4} (c+d x)} \, dx, x, x^2\right]$$

- Program code:

```
Int[1/((a_+b_.*x_^2)^(3/4)*(c_+d_.*x_^2)),x_Symbol] :=
  Sqrt[-b*x^2/a]/(2*x)*Subst[Int[1/(Sqrt[-b*x/a]*(a+b*x)^(3/4)*(c+d*x)),x],x,x^2] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

5: $\int \frac{\left(a+b x^2\right)^p}{c+d x^2} \, dx$ when $b c-a d\neq 0 \land p>0$

Derivation: Algebraic expansion

Basis: $\frac{(a+b z)^p}{c+d z} == \frac{b (a+b z)^{p-1}}{d} - \frac{(b c-a d) (a+b z)^{p-1}}{d (c+d z)}$

–

Rule 1.1.3.3.9.2.5: If $b c-a d\neq 0 \land p>0$, then

$$\int \frac{\left(a+b x^2\right)^p}{c+d x^2} \, dx \longrightarrow \frac{b}{d} \int \left(a+b x^2\right)^{p-1} \, dx - \frac{b c-a d}{d} \int \frac{\left(a+b x^2\right)^{p-1}}{c+d x^2} \, dx$$

–

Program code:

```
Int[(a_+b_.*x_^2)^p_./(c_+d_.*x_^2),x_Symbol] :=
  b/d*Int[(a+b*x^2)^(p-1),x] - (b*c-a*d)/d*Int[(a+b*x^2)^(p-1)/(c+d*x^2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && GtQ[p,0] && (EqQ[p,1/2] || EqQ[Denominator[p],4])
```

6: $\int \frac{(a+b x^2)^p}{c+d x^2} \, dx$ when $b c - a d \neq 0 \land p < -1$

Derivation: Algebraic expansion

Basis: $\frac{(a+b z)^p}{c+d z} = \frac{b (a+b z)^p}{b c-a d} - \frac{d (a+b z)^{p+1}}{(b c-a d) (c+d z)}$

Rule 1.1.3.3.9.2.6: If $b c - a d \neq 0 \land p < -1$, then

$$\int \frac{(a+b x^2)^p}{c+d x^2} \, dx \to \frac{b}{(b c-a d)} \int (a+b x^2)^p \, dx - \frac{d}{(b c-a d)} \int \frac{(a+b x^2)^{p+1}}{c+d x^2} \, dx$$

Program code:

```
Int[(a_+b_.*x_^2)^p_/(c_+d_.*x_^2),x_Symbol] :=
  b/(b*c-a*d)*Int[(a+b*x^2)^p,x] - d/(b*c-a*d)*Int[(a+b*x^2)^(p+1)/(c+d*x^2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && EqQ[Denominator[p],4] && (EqQ[p,-5/4] || EqQ[p,-7/4])
```

3. $\int \frac{(a+b x^4)^p}{c+d x^4} \, dx$ when $b c - a d \neq 0$

1. $\int \frac{(a+b x^4)^p}{c+d x^4} \, dx$ when $b c - a d \neq 0 \land p > 0$

1. $\int \frac{\sqrt{a+b x^4}}{c+d x^4} \, dx$ when $b c - a d \neq 0$

1. $\int \frac{\sqrt{a+b x^4}}{c+d x^4} \, dx$ when $b c + a d = 0$

1: $\int \frac{\sqrt{a+b x^4}}{c+d x^4} \, dx$ when $b c + a d = 0 \land a b > 0$

Derivation: Integration by substitution

Basis: If $b c + a d == 0$, then $\frac{\sqrt{a+b x^4}}{c+d x^4} == \frac{a}{c} \text{Subst}\left[\frac{1}{1-4 a b x^4}, x, \frac{x}{\sqrt{a+b x^4}}\right] \partial_x \frac{x}{\sqrt{a+b x^4}}$

Rule 1.1.3.3.9.3.1.1.1.1: If $b c + a d == 0 \land a b > 0$, then

$$\int \frac{\sqrt{a+b x^4}}{c+d x^4} dx \longrightarrow \frac{a}{c} \text{Subst}\left[\int \frac{1}{1-4 a b x^4} dx, x, \frac{x}{\sqrt{a+b x^4}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*x_^4]/(c_+d_.*x_^4),x_Symbol] :=
  a/c*Subst[Int[1/(1-4*a*b*x^4),x],x,x/Sqrt[a+b*x^4]] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && PosQ[a*b]
```

2: $\int \frac{\sqrt{a+b x^4}}{c+d x^4} dx$ when $b c + a d == 0 \land a b \ngtr 0$

Contributed by Martin Welz on 31 January 2017

Rule 1.1.3.3.9.3.1.1.1.2: If $b c + a d == 0 \land a b \ngtr 0$, let $q \to (-a b)^{1/4}$, then

$$\int \frac{\sqrt{a+b x^4}}{c+d x^4} dx \longrightarrow \frac{a}{2 c q} \text{ArcTan}\left[\frac{q x \left(a+q^2 x^2\right)}{a \sqrt{a+b x^4}}\right] + \frac{a}{2 c q} \text{ArcTanh}\left[\frac{q x \left(a-q^2 x^2\right)}{a \sqrt{a+b x^4}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*x_^4]/(c_+d_.*x_^4),x_Symbol] :=
  With[{q=Rt[-a*b,4]},
  a/(2*c*q)*ArcTan[q*x*(a+q^2*x^2)/(a*Sqrt[a+b*x^4])] + a/(2*c*q)*ArcTanh[q*x*(a-q^2*x^2)/(a*Sqrt[a+b*x^4])]] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && NegQ[a*b]
```

2: $\int \frac{\sqrt{a+b x^4}}{c+d x^4} \, dx$ when $b c - a d \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b z}}{c+d z} == \frac{b}{d \sqrt{a+b z}} - \frac{b c-a d}{d \sqrt{a+b z} \; (c+d z)}$

Rule 1.1.3.3.9.3.1.1.2: If $b c - a d \neq 0$, then

$$\int \frac{\sqrt{a+b x^4}}{c+d x^4} \, dx \longrightarrow \frac{b}{d} \int \frac{1}{\sqrt{a+b x^4}} \, dx - \frac{b c-a d}{d} \int \frac{1}{\sqrt{a+b x^4} \; (c+d x^4)} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^4]/(c_+d_.*x_^4),x_Symbol] :=
  b/d*Int[1/Sqrt[a+b*x^4],x] - (b*c-a*d)/d*Int[1/(Sqrt[a+b*x^4]*(c+d*x^4)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

2: $\int \frac{(a+b x^4)^{1/4}}{c+d x^4} \, dx$ when $b c - a d \neq 0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \sqrt{a+b x^4} \sqrt{\frac{a}{a+b x^4}} == 0$

Basis: $\frac{1}{\sqrt{\frac{a}{a+b x^4}} \; (a+b x^4)^{1/4} \; (c+d x^4)} == \text{Subst}\left[\frac{1}{\sqrt{1-b x^4} \; (c-(b c-a d) x^4)}, x, \frac{x}{(a+b x^4)^{1/4}}\right] \partial_x \frac{x}{(a+b x^4)^{1/4}}$

Rule 1.1.3.3.9.3.1.2: If $b c - a d \neq 0$, then

$$\int \frac{\left(a+b x^4\right)^{1/4}}{c+d x^4}\,dx \to \sqrt{a+b x^4}\,\sqrt{\frac{a}{a+b x^4}}\int \frac{1}{\sqrt{\frac{a}{a+b x^4}}\,\left(a+b x^4\right)^{1/4}\left(c+d x^4\right)}\,dx$$

$$\to \sqrt{a+b x^4}\,\sqrt{\frac{a}{a+b x^4}}\,\text{Subst}\left[\int \frac{1}{\sqrt{1-b x^4}\,\left(c-(b c-a d)\,x^4\right)}\,dx,\,x,\,\frac{x}{\left(a+b x^4\right)^{1/4}}\right]$$

Program code:

```
Int[(a_+b_.*x_^4)^(1/4)/(c_+d_.*x_^4),x_Symbol] :=
  Sqrt[a+b*x^4]*Sqrt[a/(a+b*x^4)]*Subst[Int[1/(Sqrt[1-b*x^4]*(c-(b*c-a*d)*x^4)),x],x,x/(a+b*x^4)^(1/4)] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

3: $\int \frac{\left(a+b x^4\right)^{5/4}}{c+d x^4}\,dx$ when $b c-a d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{(a+b z)^p}{c+d z}==\frac{b\,(a+b z)^{p-1}}{d}-\frac{(b c-a d)\,(a+b z)^{p-1}}{d\,(c+d z)}$

Rule 1.1.3.3.9.3.1.3: If $b c-a d\neq 0$, then

$$\int \frac{\left(a+b x^4\right)^{5/4}}{c+d x^4}\,dx \to \frac{b}{d}\int \left(a+b x^4\right)^{1/4}dx-\frac{b c-a d}{d}\int \frac{\left(a+b x^4\right)^{1/4}}{c+d x^4}\,dx$$

Program code:

```
Int[(a_+b_.*x_^4)^(5/4)/(c_+d_.*x_^4),x_Symbol] :=
  b/d*Int[(a+b*x^4)^(1/4),x] - (b*c-a*d)/d*Int[(a+b*x^4)^(1/4)/(c+d*x^4),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

2. $\int \frac{\left(a+b x^4\right)^p}{c+d x^4}\,dx$ when $b c-a d\neq 0\land p<0$

1: $\int \frac{1}{\sqrt{a+b x^4}\,\left(c+d x^4\right)}\,dx$ when $b c-a d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{c+d x^4}==\frac{1}{2 c\left(1-\sqrt{-\frac{d}{c}}\;x^2\right)}+\frac{1}{2 c\left(1+\sqrt{-\frac{d}{c}}\;x^2\right)}$

Rule 1.1.3.3.9.3.2.1: If $b c-a d\neq 0$, then

$$\int \frac{1}{\sqrt{a+b x^4}\,\left(c+d x^4\right)}\,dx\to\frac{1}{2 c}\int \frac{1}{\sqrt{a+b x^4}\,\left(1-\sqrt{-\frac{d}{c}}\;x^2\right)}\,dx+\frac{1}{2 c}\int \frac{1}{\sqrt{a+b x^4}\,\left(1+\sqrt{-\frac{d}{c}}\;x^2\right)}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_^4]*(c_+d_.*x_^4)),x_Symbol] :=
  1/(2*c)*Int[1/(Sqrt[a+b*x^4]*(1-Rt[-d/c,2]*x^2)),x] + 1/(2*c)*Int[1/(Sqrt[a+b*x^4]*(1+Rt[-d/c,2]*x^2)),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

2: $\int \frac{1}{\left(a+b x^4\right)^{3/4} \left(c+d x^4\right)} \, dx$ when $b c - a d \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{(a+b z)^p}{c+d z} = \frac{b (a+b z)^p}{b c-a d} - \frac{d (a+b z)^{p+1}}{(b c-a d) (c+d z)}$

Rule 1.1.3.3.9.3.2.2: If $b c - a d \neq 0$, then

$$\int \frac{1}{\left(a+b x^4\right)^{3/4} \left(c+d x^4\right)} \, dx \longrightarrow \frac{b}{(b c - a d)} \int \frac{1}{\left(a+b x^4\right)^{3/4}} \, dx - \frac{d}{(b c - a d)} \int \frac{\left(a+b x^4\right)^{1/4}}{c+d x^4} \, dx$$

Program code:

```
Int[1/((a_+b_.*x_^4)^(3/4)*(c_+d_.*x_^4)),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(a+b*x^4)^(3/4),x] - d/(b*c-a*d)*Int[(a+b*x^4)^(1/4)/(c+d*x^4),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

4. $\int \frac{(a+b\,x^3)^p}{c+d\,x^3}\,dx$ when $b\,c-a\,d\neq 0\wedge b\,c+a\,d==0\wedge p-\frac{1}{3}\in\mathbb{Z}$

1: $\int \frac{(a+b\,x^3)^{1/3}}{c+d\,x^3}\,dx$ when $b\,c-a\,d\neq 0\wedge b\,c+a\,d==0$

Derivation: Integration by substitution

Basis: If $b\,c+a\,d==0$, let $q\to\left(\frac{b}{a}\right)^{1/3}$, then $\frac{(a+b\,x^3)^{1/3}}{c+d\,x^3}==\frac{9\,a}{c\,q}\,\text{Subst}\left[\frac{x}{(4-a\,x^3)\,(1+2\,a\,x^3)},\,x,\,\frac{(1+q\,x)}{(a+b\,x^3)^{1/3}}\right]\partial_x\frac{(1+q\,x)}{(a+b\,x^3)^{1/3}}$

Rule 1.1.3.3.9.4.1: If $b\,c-a\,d\neq 0\wedge b\,c+a\,d==0$, let $q\to\left(\frac{b}{a}\right)^{1/3}$, then

$$\int\frac{(a+b\,x^3)^{1/3}}{c+d\,x^3}\,dx\;\longrightarrow\;\frac{9\,a}{c\,q}\,\text{Subst}\left[\int\frac{x}{(4-a\,x^3)\,(1+2\,a\,x^3)}\,dx,\,x,\,\frac{(1+q\,x)}{(a+b\,x^3)^{1/3}}\right]$$

Program code:

```
Int[(a_+b_.*x_^3)^(1/3)/(c_+d_.*x_^3),x_Symbol] :=
  With[{q=Rt[b/a,3]},
  9*a/(c*q)*Subst[Int[x/((4-a*x^3)*(1+2*a*x^3)),x],x,(1+q*x)/(a+b*x^3)^(1/3)]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b*c+a*d,0]
```

2: $\int \frac{1}{\left(a+b x^3\right)^{2/3} \left(c+d x^3\right)} \, dx$ when $b c-a d\neq 0 \land b c+a d=0$

Derivation: Algebraic expansion

Basis: $(a+b z)^p (c+d z)^q = \frac{b (a+b z)^p (c+d z)^{q+1}}{b c-a d} - \frac{d (a+b z)^{p+1} (c+d z)^q}{b c-a d}$

Rule 1.1.3.3.9.4.2: If $b c-a d\neq 0 \land b c+a d=0$, then

$$\int \frac{1}{\left(a+b x^3\right)^{2/3} \left(c+d x^3\right)} \, dx \to \frac{b}{b c-a d} \int \frac{1}{\left(a+b x^3\right)^{2/3}} \, dx - \frac{d}{b c-a d} \int \frac{\left(a+b x^3\right)^{1/3}}{c+d x^3} \, dx$$

Program code:

```
Int[1/((a_+b_.*x_^3)^(2/3)*(c_+d_.*x_^3)),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(a+b*x^3)^(2/3),x] - d/(b*c-a*d)*Int[(a+b*x^3)^(1/3)/(c+d*x^3),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b*c+a*d,0]
```

10. $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c-a d\neq 0 \land p<-1$

1. $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c-a d\neq 0 \land p<-1 \land q>0$

1: $\int \frac{\sqrt{a+b x^2}}{\left(c+d x^2\right)^{3/2}} \, dx$ when $\frac{b}{a}>0 \land \frac{d}{c}>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2} \sqrt{\frac{c (a+b x^2)}{a (c+d x^2)}}} = 0$

Rule 1.1.3.3.10.1.1: If $\frac{b}{a}>0 \land \frac{d}{c}>0$, then

$$\int \frac{\sqrt{a+b x^2}}{\left(c+d x^2\right)^{3/2}}\,dx \longrightarrow \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}\int \frac{\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{c+d x^2}\,dx \longrightarrow \frac{\sqrt{a+b x^2}}{c\,\sqrt{\frac{d}{c}}\,\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}\,\text{EllipticE}\!\left[\text{ArcTan}\!\left[\sqrt{\frac{d}{c}}\,x\right],\,1-\frac{b\,c}{a\,d}\right]$$

$$\int \frac{\sqrt{a+b x^2}}{\left(c+d x^2\right)^{3/2}}\,dx \longrightarrow \frac{a\,\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{c\,\sqrt{a+b x^2}}\int \frac{\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{c+d x^2}\,dx \longrightarrow \frac{a\,\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{c^2\,\sqrt{\frac{d}{c}}\,\sqrt{a+b x^2}}\,\text{EllipticE}\!\left[\text{ArcTan}\!\left[\sqrt{\frac{d}{c}}\,x\right],\,1-\frac{b\,c}{a\,d}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  Sqrt[a+b*x^2]/(c*Rt[d/c,2]*Sqrt[c+d*x^2]*Sqrt[c*(a+b*x^2)/(a*(c+d*x^2))])*EllipticE[ArcTan[Rt[d/c,2]*x],1-b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && PosQ[b/a] && PosQ[d/c]
```

```
(* Int[Sqrt[a_+b_.*x_^2]/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  a*Sqrt[c+d*x^2]*Sqrt[(c*(a+b*x^2))/(a*(c+d*x^2))]/(c^2*Rt[d/c,2]*Sqrt[a+b*x^2])*EllipticE[ArcTan[Rt[d/c,2]*x],1-b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && PosQ[b/a] && PosQ[d/c] *)
```

2: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land p < -1 \land 0 < q < 1$

Derivation: Binomial product recurrence 1 with $A = 1$ and $B = 0$

Rule 1.1.3.3.10.1.2: If $b c - a d \neq 0 \land p < -1 \land 0 < q < 1$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow -\frac{x (a+b x^n)^{p+1} (c+d x^n)^q}{a n (p+1)} + \frac{1}{a n (p+1)} \int (a+b x^n)^{p+1} (c+d x^n)^{q-1} \left(c (n (p+1)+1) + d (n (p+q+1)+1) x^n\right) dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -x*(a+b*x^n)^(p+1)*(c+d*x^n)^q/(a*n*(p+1)) +
  1/(a*n*(p+1))*Int[(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)*Simp[c*(n*(p+1)+1)+d*(n*(p+q+1)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && LtQ[0,q,1] && IntBinomialQ[a,b,c,d,n,p,q,x]
```

3: $\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$ when $b\,c-a\,d\neq 0\land p<-1\land q>1$

Derivation: Binomial product recurrence 1 with $A = c$, $B = d$ and $q = q-1$

Rule 1.1.3.3.10.1.3: If $b\,c-a\,d\neq 0\land p<-1\land q>1$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx \to \frac{(a\,d-b\,c)\,x\,(a+b\,x^n)^{p+1}\,(c+d\,x^n)^{q-1}}{a\,b\,n\,(p+1)} - \frac{1}{a\,b\,n\,(p+1)}\int (a+b\,x^n)^{p+1}\,(c+d\,x^n)^{q-2}\,(c\,(a\,d-b\,c\,(n\,(p+1)+1))+d\,(a\,d\,(n\,(q-1)+1)-b\,c\,(n\,(p+q)+1))\,x^n)\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  (a*d-c*b)*x*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)/(a*b*n*(p+1)) -
  1/(a*b*n*(p+1))*
    Int[(a+b*x^n)^(p+1)*(c+d*x^n)^(q-2)*Simp[c*(a*d-c*b*(n*(p+1)+1))+d*(a*d*(n*(q-1)+1)-b*c*(n*(p+q)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && GtQ[q,1] && IntBinomialQ[a,b,c,d,n,p,q,x]
```

2: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \wedge p < -1$

Derivation: Binomial product recurrence 2a with $A = 1$ and $B = 0$

Rule 1.1.3.3.10.1.2: If $b c - a d \neq 0 \wedge p < -1$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow -\frac{b x (a+b x^n)^{p+1} (c+d x^n)^{q+1}}{a n (p+1) (b c - a d)} + \frac{1}{a n (p+1) (b c - a d)} \int (a+b x^n)^{p+1} (c+d x^n)^q \left(b c + n (p+1) (b c - a d) + d b (n (p+q+2) + 1) x^n\right) dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  -b*x*(a+b*x^n)^(p+1)*(c+d*x^n)^(q+1)/(a*n*(p+1)*(b*c-a*d)) +
  1/(a*n*(p+1)*(b*c-a*d))*
    Int[(a+b*x^n)^(p+1)*(c+d*x^n)^q*Simp[b*c+n*(p+1)*(b*c-a*d)+d*b*(n*(p+q+2)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,n,q},x] && NeQ[b*c-a*d,0] && LtQ[p,-1] && Not[Not[IntegerQ[p]] && IntegerQ[q] && LtQ[q,-1]] &&
  IntBinomialQ[a,b,c,d,n,p,q,x]
```

11: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c-a d \neq 0 \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z} \land q \in \mathbb{Z} \land p+q>0$

– Derivation: Algebraic expansion

– Rule 1.1.3.3.11: If $b c-a d \neq 0 \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z} \land q \in \mathbb{Z} \land p+q>0$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(a+b x^n)^p (c+d x^n)^q, x\right] dx$$

– Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n)^p*(c+d*x^n)^q,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && IGtQ[n,0] && IntegersQ[p,q] && GtQ[p+q,0]
```

12. $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land q > 0$

1: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land q > 1 \land n (p+q) + 1 \neq 0$

Derivation: Binomial product recurrence 3a with $A = c$, $B = d$ and $q = q - 1$

Rule 1.1.3.3.12.1: If $b c - a d \neq 0 \land q > 1 \land n (p+q) + 1 \neq 0$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow \frac{d x (a+b x^n)^{p+1} (c+d x^n)^{q-1}}{b (n (p+q) + 1)} +$$
$$\frac{1}{b (n (p+q) + 1)} \int (a+b x^n)^p (c+d x^n)^{q-2} \left(c (b c (n (p+q) + 1) - a d) + d (b c (n (p+2q-1) + 1) - a d (n (q-1) + 1)) x^n\right) dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  d*x*(a+b*x^n)^(p+1)*(c+d*x^n)^(q-1)/(b*(n*(p+q)+1)) +
  1/(b*(n*(p+q)+1))*
    Int[(a+b*x^n)^p*(c+d*x^n)^(q-2)*Simp[c*(b*c*(n*(p+q)+1)-a*d)+d*(b*c*(n*(p+2*q-1)+1)-a*d*(n*(q-1)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d,n,p},x] && NeQ[b*c-a*d,0] && GtQ[q,1] && NeQ[n*(p+q)+1,0] && Not[IGtQ[p,1]] && IntBinomialQ[a,b,c,d,n,p,q,x]
```

2: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land q > 0 \land p > 0$

Derivation: Binomial product recurrence 2b with $m = 0$, $A = a$, $B = b$ and $p = p - 1$

Rule 1.1.3.3.12.2: If $b c - a d \neq 0 \land q > 0 \land p > 0$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow$$
$$\frac{x (a+b x^n)^p (c+d x^n)^q}{n (p+q)+1} + \frac{n}{n (p+q)+1} \int (a+b x^n)^{p-1} (c+d x^n)^{q-1} \left(a c (p+q) + (q (b c - a d) + a d (p+q)) x^n\right) dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  x*(a+b*x^n)^p*(c+d*x^n)^q/(n*(p+q)+1) +
  n/(n*(p+q)+1)*Int[(a+b*x^n)^(p-1)*(c+d*x^n)^(q-1)*Simp[a*c*(p+q)+(q*(b*c-a*d)+a*d*(p+q))*x^n,x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && GtQ[q,0] && GtQ[p,0] && IntBinomialQ[a,b,c,d,n,p,q,x]
```

13. $\int \frac{(a+b x^2)^p}{\sqrt{c+d x^2}} \, dx$ when $b c - a d \neq 0 \land p^2 == \frac{1}{4}$

1. $\int \frac{1}{\sqrt{a+b x^2} \sqrt{c+d x^2}} \, dx$ when $b c - a d \neq 0$

1: $\int \frac{1}{\sqrt{a+b x^2} \sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} > 0 \land \frac{b}{a} > 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{c+d x^2} \sqrt{\frac{c (a+b x^2)}{a (c+d x^2)}}}{\sqrt{a+b x^2}} == 0$

Rule 1.1.3.3.13.1.1: If $\frac{d}{c} > 0 \land \frac{b}{a} > 0$, then

$$\int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}\,dx \longrightarrow \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}\int \frac{\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{a+b x^2}\,dx \longrightarrow \frac{\sqrt{a+b x^2}}{a\,\sqrt{\frac{d}{c}}\,\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}\,\text{EllipticF}\left[\text{ArcTan}\left[\sqrt{\frac{d}{c}}\,x\right],\,1-\frac{b c}{a d}\right]$$

$$\int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}\,dx \longrightarrow \frac{a\,\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{c\,\sqrt{a+b x^2}}\int \frac{\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{a+b x^2}\,dx \longrightarrow \frac{\sqrt{c+d x^2}\,\sqrt{\frac{c\,(a+b x^2)}{a\,(c+d x^2)}}}{c\,\sqrt{\frac{d}{c}}\,\sqrt{a+b x^2}}\,\text{EllipticF}\left[\text{ArcTan}\left[\sqrt{\frac{d}{c}}\,x\right],\,1-\frac{b c}{a d}\right]$$

–

Program code:

```
Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Sqrt[a+b*x^2]/(a*Rt[d/c,2]*Sqrt[c+d*x^2]*Sqrt[c*(a+b*x^2)/(a*(c+d*x^2))])*EllipticF[ArcTan[Rt[d/c,2]*x],1-b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && PosQ[d/c] && PosQ[b/a] && Not[SimplerSqrtQ[b/a,d/c]]
```

```
(* Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Sqrt[c+d*x^2]*Sqrt[c*(a+b*x^2)/(a*(c+d*x^2))]/(c*Rt[d/c,2]*Sqrt[a+b*x^2])*EllipticF[ArcTan[Rt[d/c,2]*x],1-b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && PosQ[d/c] && PosQ[b/a] && Not[SimplerSqrtQ[b/a,d/c]] *)
```

2. $\int \frac{1}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}\,dx$ when $\frac{d}{c} \ngtr 0$

1: $\int \frac{1}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}\,dx$ when $\frac{d}{c} \ngtr 0 \land c>0 \land a>0$

–

Rule 1.1.3.3.13.1.2.1: If $\frac{d}{c} \ngtr 0 \land c>0 \land a>0$, then

$$\int \frac{1}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}\,dx \to \frac{1}{\sqrt{a}\ \sqrt{c}\ \sqrt{-\frac{d}{c}}}\ \text{EllipticF}\left[\text{ArcSin}\left[\sqrt{-\frac{d}{c}}\ x\right],\ \frac{b c}{a d}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  1/(Sqrt[a]*Sqrt[c]*Rt[-d/c,2])*EllipticF[ArcSin[Rt[-d/c,2]*x],b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && NegQ[d/c] && GtQ[c,0] && GtQ[a,0] && Not[NegQ[b/a] && SimplerSqrtQ[-b/a,-d/c]]
```

2: $\int \frac{1}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}\,dx$ when $\frac{d}{c} \ngtr 0 \land c>0 \land a-\frac{b c}{d}>0$

–

Rule 1.1.3.3.13.1.2.2: If $\frac{d}{c} \ngtr 0 \land c>0 \land a-\frac{b c}{d}>0$, then

$$\int \frac{1}{\sqrt{a+b x^2}\ \sqrt{c+d x^2}}\,dx \to -\frac{1}{\sqrt{c}\ \sqrt{-\frac{d}{c}}\ \sqrt{a-\frac{b c}{d}}}\ \text{EllipticF}\left[\text{ArcCos}\left[\sqrt{-\frac{d}{c}}\ x\right],\ \frac{b c}{b c-a d}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  -1/(Sqrt[c]*Rt[-d/c,2]*Sqrt[a-b*c/d])*EllipticF[ArcCos[Rt[-d/c,2]*x],b*c/(b*c-a*d)] /;
FreeQ[{a,b,c,d},x] && NegQ[d/c] && GtQ[c,0] && GtQ[a-b*c/d,0]
```

3: $\int \frac{1}{\sqrt{a+b x^2} \sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} \not> 0 \land c \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{d}{c} x^2}}{\sqrt{c+d x^2}} = 0$

Rule 1.1.3.3.13.1.2.3: If $\frac{d}{c} \not> 0 \land c \not> 0$, then

$$\int \frac{1}{\sqrt{a+b x^2} \sqrt{c+d x^2}} \, dx \to \frac{\sqrt{1+\frac{d}{c} x^2}}{\sqrt{c+d x^2}} \int \frac{1}{\sqrt{a+b x^2} \sqrt{1+\frac{d}{c} x^2}} \, dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Sqrt[1+d/c*x^2]/Sqrt[c+d*x^2]*Int[1/(Sqrt[a+b*x^2]*Sqrt[1+d/c*x^2]),x] /;
FreeQ[{a,b,c,d},x] && Not[GtQ[c,0]]
```

2. $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx$ when $b c - a d \neq 0$

1. $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} > 0$

1: $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} > 0 \land \frac{b}{a} > 0$

Derivation: Algebraic expansion

Basis: $\sqrt{a+b x^2} == \frac{a}{\sqrt{a+b x^2}} + \frac{b x^2}{\sqrt{a+b x^2}}$

Rule 1.1.3.3.13.2.1.1: If $\frac{d}{c} > 0 \land \frac{b}{a} > 0$, then

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx \longrightarrow a \int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}} \, dx + b \int \frac{x^2}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  a*Int[1/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x] + b*Int[x^2/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x] /;
FreeQ[{a,b,c,d},x] && PosQ[d/c] && PosQ[b/a]
```

2: $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} > 0 \land \frac{b}{a} \ngtr 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} == \frac{b \sqrt{c+d x^2}}{d \sqrt{a+b x^2}} - \frac{b c-a d}{d \sqrt{a+b x^2}\,\sqrt{c+d x^2}}$

Rule 1.1.3.3.13.2.1.2: If $\frac{d}{c} > 0 \land \frac{b}{a} \ngtr 0$, then

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx \longrightarrow \frac{b}{d} \int \frac{\sqrt{c+d x^2}}{\sqrt{a+b x^2}} \, dx - \frac{b c-a d}{d} \int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  b/d*Int[Sqrt[c+d*x^2]/Sqrt[a+b*x^2],x] - (b*c-a*d)/d*Int[1/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x] /;
FreeQ[{a,b,c,d},x] && PosQ[d/c] && NegQ[b/a]
```

2. $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} \ngtr 0$

1. $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} \ngtr 0 \wedge c > 0$

1: $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx$ when $\frac{d}{c} \ngtr 0 \wedge c > 0 \wedge a > 0$

Rule 1.1.3.3.13.2.2.1.1: If $\frac{d}{c} \ngtr 0 \wedge c > 0 \wedge a > 0$, then

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx \to \frac{\sqrt{a}}{\sqrt{c} \sqrt{-\frac{d}{c}}} \text{EllipticE}\left[\text{ArcSin}\left[\sqrt{-\frac{d}{c}} \, x\right], \frac{b c}{a d}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  Sqrt[a]/(Sqrt[c]*Rt[-d/c,2])*EllipticE[ArcSin[Rt[-d/c,2]*x],b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && NegQ[d/c] && GtQ[c,0] && GtQ[a,0]
```

2: $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}}\,dx$ when $\frac{d}{c} \ngtr 0 \land c>0 \land a-\frac{b c}{d}>0$

- Rule 1.1.3.3.13.2.2.1.2: If $\frac{d}{c} \ngtr 0 \land c>0 \land a-\frac{b c}{d}>0$, then

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}}\,dx \to -\frac{\sqrt{a-\frac{b c}{d}}}{\sqrt{c}\,\sqrt{-\frac{d}{c}}}\,\text{EllipticE}\left[\text{ArcCos}\left[\sqrt{-\frac{d}{c}}\,x\right],\frac{b c}{b c-a d}\right]$$

- Program code:

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  -Sqrt[a-b*c/d]/(Sqrt[c]*Rt[-d/c,2])*EllipticE[ArcCos[Rt[-d/c,2]*x],b*c/(b*c-a*d)] /;
FreeQ[{a,b,c,d},x] && NegQ[d/c] && GtQ[c,0] && GtQ[a-b*c/d,0]
```

3: $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}}\, dx$ when $\frac{d}{c} \ngtr 0 \land c > 0 \land a \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a+b x^2}}{\sqrt{1+\frac{b}{a} x^2}} == 0$

Rule 1.1.3.3.13.2.2.1.3: If $\frac{d}{c} \ngtr 0 \land c > 0 \land a \ngtr 0$, then

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}}\, dx \longrightarrow \frac{\sqrt{a+b x^2}}{\sqrt{1+\frac{b}{a} x^2}} \int \frac{\sqrt{1+\frac{b}{a} x^2}}{\sqrt{c+d x^2}}\, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  Sqrt[a+b*x^2]/Sqrt[1+b/a*x^2]*Int[Sqrt[1+b/a*x^2]/Sqrt[c+d*x^2],x] /;
FreeQ[{a,b,c,d},x] && NegQ[d/c] && GtQ[c,0] && Not[GtQ[a,0]]
```

2: $\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}}\, dx$ when $\frac{d}{c} \ngtr 0 \land c \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{d}{c} x^2}}{\sqrt{c+d x^2}} == 0$

Rule 1.1.3.3.13.2.2.2: If $\frac{d}{c} \ngtr 0 \land c \ngtr 0$, then

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx \to \frac{\sqrt{1+\frac{d}{c} x^2}}{\sqrt{c+d x^2}} \int \frac{\sqrt{a+b x^2}}{\sqrt{1+\frac{d}{c} x^2}} \, dx$$

–

Program code:

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  Sqrt[1+d/c*x^2]/Sqrt[c+d*x^2]*Int[Sqrt[a+b*x^2]/Sqrt[1+d/c*x^2],x] /;
FreeQ[{a,b,c,d},x] && NegQ[d/c] && Not[GtQ[c,0]]
```

14: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c - a d \neq 0 \land p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.3.14: If $b c - a d \neq 0 \land p \in \mathbb{Z}^+$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \to \int \text{ExpandIntegrand}\left[(a+b x^n)^p (c+d x^n)^q, x\right] dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_.*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x^n)^p*(c+d*x^n)^q,x],x] /;
FreeQ[{a,b,c,d,n,q},x] && NeQ[b*c-a*d,0] && IGtQ[p,0]
```

A. $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c-a d \neq 0 \land n \neq -1$

1: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c-a d \neq 0 \land n \neq -1 \land (p \in \mathbb{Z} \lor a>0) \land (q \in \mathbb{Z} \lor c>0)$

– Rule 1.1.3.3.A.1: If $b c-a d \neq 0 \land n \neq -1 \land (p \in \mathbb{Z} \lor a>0) \land (q \in \mathbb{Z} \lor c>0)$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow a^p c^q x \,\text{AppellF1}\left[\frac{1}{n}, -p, -q, 1+\frac{1}{n}, -\frac{b x^n}{a}, -\frac{d x^n}{c}\right]$$

– Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  a^p*c^q*x*AppellF1[1/n,-p,-q,1+1/n,-b*x^n/a,-d*x^n/c] /;
FreeQ[{a,b,c,d,n,p,q},x] && NeQ[b*c-a*d,0] && NeQ[n,-1] && (IntegerQ[p] || GtQ[a,0]) && (IntegerQ[q] || GtQ[c,0])
```

2: $\int (a+b x^n)^p (c+d x^n)^q \, dx$ when $b c-a d \neq 0 \land n \neq -1 \land \neg (p \in \mathbb{Z} \lor a>0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+b x^n)^p}{\left(1+\frac{b x^n}{a}\right)^p} == 0$

– Rule 1.1.3.3.A.2: If $b c-a d \neq 0 \land n \neq -1 \land \neg (p \in \mathbb{Z} \lor a>0)$, then

$$\int (a+b x^n)^p (c+d x^n)^q \, dx \longrightarrow \frac{a^{\text{IntPart}[p]} (a+b x^n)^{\text{FracPart}[p]}}{\left(1+\frac{b x^n}{a}\right)^{\text{FracPart}[p]}} \int \left(1+\frac{b x^n}{a}\right)^p (c+d x^n)^q \, dx$$

– Program code:

```
Int[(a_+b_.*x_^n_)^p_*(c_+d_.*x_^n_)^q_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^n)^FracPart[p]/(1+b*x^n/a)^FracPart[p]*Int[(1+b*x^n/a)^p*(c+d*x^n)^q,x] /;
FreeQ[{a,b,c,d,n,p,q},x] && NeQ[b*c-a*d,0] && NeQ[n,-1] && Not[IntegerQ[p] || GtQ[a,0]]
```

S: $\int (a+b\,u^n)^p\,(c+d\,u^n)^q\,dx$ when $u == e+f\,x$

Derivation: Integration by substitution

Rule 1.1.3.3.S: If $u == e+f\,x$, then

$$\int (a+b\,u^n)^p\,(c+d\,u^n)^q\,dx \to \frac{1}{f}\,\text{Subst}\left[\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx,\,x,\,u\right]$$

Program code:

```
Int[(a_.+b_.*u_^n_)^p_.*(c_.+d_.*u_^n_)^q_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x^n)^p*(c+d*x^n)^q,x],x,u] /;
FreeQ[{a,b,c,d,n,p,q},x] && LinearQ[u,x] && NeQ[u,x]
```

N: $\int P_x{}^p\,Q_x{}^q\,dx$ when $P_x == a+b\,(e+f\,x)^n \land Q_x == c+d\,(e+f\,x)^n$

Derivation: Algebraic normalization

Rule 1.1.3.3.N: If $P_x == a+b\,(e+f\,x)^n \land Q_x == c+d\,(e+f\,x)^n$, then

$$\int P_x{}^p\,Q_x{}^q\,dx \to \int \left(a+b\,(e+f\,x)^n\right)^p\,\left(c+d\,(e+f\,x)^n\right)^q\,dx$$

Program code:

```
Int[u_^p_.*v_^q_.,x_Symbol] :=
  Int[NormalizePseudoBinomial[u,x]^p*NormalizePseudoBinomial[v,x]^q,x] /;
FreeQ[{p,q},x] && PseudoBinomialPairQ[u,v,x]
```

```
Int[x_^m_.*u_^p_.*v_^q_.,x_Symbol] :=
  Int[NormalizePseudoBinomial[x^(m/p)*u,x]^p*NormalizePseudoBinomial[v,x]^q,x] /;
FreeQ[{p,q},x] && IntegersQ[p,m/p] && PseudoBinomialPairQ[x^(m/p)*u,v,x]
```

```
(* IntBinomialQ[a,b,c,d,n,p,q,x] returns True iff (a+b*x^n)^p*(c+d*x^n)^q is integrable wrt x in terms of non-Appell functions. *)
IntBinomialQ[a_,b_,c_,d_,n_,p_,q_,x_Symbol] :=
  IntegersQ[p,q] || IGtQ[p,0] || IGtQ[q,0] ||
  (EqQ[n,2] || EqQ[n,4]) && (IntegersQ[p,4*q] || IntegersQ[4*p,q]) ||
  EqQ[n,2] && (IntegersQ[2*p,2*q] || IntegersQ[3*p,q] && EqQ[b*c+3*a*d,0] || IntegersQ[p,3*q] && EqQ[3*b*c+a*d,0]) ||
  EqQ[n,3] && (IntegersQ[p+1/3,q] || IntegersQ[q+1/3,p]) ||
  EqQ[n,3] && (IntegersQ[p+2/3,q] || IntegersQ[q+2/3,p]) && EqQ[b*c+a*d,0]
```

Rules for integrands of the form $(a+b\,x^n)^p\,(c+d\,x^{-n})^q$

1: $\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,dx$ when $q\in\mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $q\in\mathbb{Z}$, then $(c+d\,x^{-n})^q == \frac{(d+c\,x^n)^q}{x^{n\,q}}$

Rule 1.1.3.3.15.1: If $q\in\mathbb{Z}$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,dx \longrightarrow \int \frac{(a+b\,x^n)^p\,(d+c\,x^n)^q}{x^{n\,q}}\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^mn_.)^q_.,x_Symbol] :=
  Int[(a+b*x^n)^p*(d+c*x^n)^q/x^(n*q),x] /;
FreeQ[{a,b,c,d,n,p},x] && EqQ[mn,-n] && IntegerQ[q] && (PosQ[n] || Not[IntegerQ[p]])
```

2: $\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,dx$ when $q\notin\mathbb{Z}\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{n\,q}\,(c+d\,x^{-n})^q}{(d+c\,x^n)^q} == 0$

Basis: $\frac{x^{n\,q}\,(c+d\,x^{-n})^q}{(d+c\,x^n)^q} == \frac{x^{n\,\mathrm{FracPart}[q]}\,(c+d\,x^{-n})^{\mathrm{FracPart}[q]}}{(d+c\,x^n)^{\mathrm{FracPart}[q]}}$

Rule 1.1.3.3.15.2: If $q\notin\mathbb{Z}\,\wedge\,p\notin\mathbb{Z}$, then

$$\int (a+b\,x^n)^p\,(c+d\,x^{-n})^q\,dx \longrightarrow \frac{x^{n\,\mathrm{FracPart}[q]}\,(c+d\,x^{-n})^{\mathrm{FracPart}[q]}}{(d+c\,x^n)^{\mathrm{FracPart}[q]}}\int \frac{(a+b\,x^n)^p\,(d+c\,x^n)^q}{x^{n\,q}}\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_.)^p_*(c_+d_.*x_^mn_.)^q_,x_Symbol] :=
  x^(n*FracPart[q])*(c+d*x^(-n))^FracPart[q]/(d+c*x^n)^FracPart[q]*Int[(a+b*x^n)^p*(d+c*x^n)^q/x^(n*q),x] /;
FreeQ[{a,b,c,d,n,p,q},x] && EqQ[mn,-n] && Not[IntegerQ[q]] && Not[IntegerQ[p]]
```